Encapsulating File Access in a Service Program 

by Joel Cochran 

Encapsulation is a familiar concept to object-oriented programmers but has yet to be seriously embraced by RPG programmers. The idea is to place all access for a particular set of data, in this case from a physical file, to one object. Then only one object can directly access and manipulate a database file. While this is standard fare for OO languages, it is hardly the norm for RPG. In this article I demonstrate a method that simulates encapsulation by using our trusty friend the service program. 

WHY ENCAPSULATION IS COOL 

Let's start off with a simple file of names and e-mail addresses based on the following DDS: 

A                                      UNIQUE

A          R MYNAMER                   TEXT('NAMES AND EMAILS ADDR

A            ID                        5S 0       COLHDG('ID KEY')

A            USERNAME      35A         COLHDG('USER NAME')

A            USEREMAIL    128A         COLHDG('USER EMAIL ADDRESS'

A          K ID

Now assume that through years of maintenance and development our little file here has been used in over a hundred programs. After so much propagation, you need to verify that the format of the e-mail address is correct. User error causes problems that you can limit if you control how the data is entering the file. 

This usually would mean finding all instances of updates and writes and adding validation logic in each place. How many times have you seen a subroutine like this copied all over the place? But since we are more modern RPG programmers, we could easily enough write a validateEmail() routine and stick it in a service program. This is probably what we should do, but it doesn't change the fact that we still need to find all the instances that update or write to the file and add the call to the new subprocedure. 

Another typical problem is adding a new field to the file, say a telephone number. It's easy enough with DDS: 

A                                      UNIQUE

A          R MYNAMER                   TEXT('NAMES AND EMAILS ADDR

A            ID                        5S 0       COLHDG('ID KEY')

A            USERNAME      35A         COLHDG('USER NAME')

A            USEREMAIL    128A         COLHDG('USER EMAIL ADDRESS'

A            PHONENUM      10S 0    COLHDG('USER PHONE'

A          K ID

As you probably already recognize, the problem here is the level check. You now have to find and recompile every program that accesses the file in order to prevent a runtime error. In a lot of software shops, database changes like this to legacy applications are practically forbidden because of such problems. In both examples, a lot of additional work has to be done to ensure the application keeps running without errors. 

Now imagine if you had used encapsulation for this file. Either adding a field or instituting some rules would require changing and recompiling only a single source member; the magic of UPDSRVPGM would take care of the rest. From a maintenance perspective, that is very cool. 

Another cool thing about encapsulation is that you can hide a lot of the nitty gritty details. If you write a service program for other programmers to use, you can minimize the amount of work they have to do in order to properly use the file, while allowing you to easily enforce your rules. Let's add two more fields to the file above that represent the user profile that updated the record, and a timestamp: 

A                                      UNIQUE

A          R MYNAMER                   TEXT('NAMES AND EMAILS ADDR

A            ID                        5S 0       COLHDG('ID KEY')

A            USERNAME      35A         COLHDG('USER NAME')

A            USEREMAIL    128A         COLHDG('USER EMAIL ADDRESS'

A            PHONENUM      10S 0       COLHDG('USER PHONE')

A            UPDUSER          10A         COLHDG('UPDATE USER')

A            UPDSTAMP           Z         COLHDG('UPDATE TIMESTAMP')

A          K ID

Again, if you really wanted to enforce this, you would have to find all the write and update statements for this file and add these fields to the program logic. With encapsulation you only need to add this in one place. Even more appealing is that the programmer using your service program doesn't even have to code for it: the encapsulating service program can make this an essentially automatic feature. I'll demonstrate this later in the article. 

THE PIECES OF THE PIE 

There are some standard procedures that need to be included in your encapsulating service program. These procedures break down into two categories. First, you need file access procedures for reading, updating, inserting, and deleting. Second, you need accessor procedures for the individual fields. Because your file access is hidden within your encapsulating service program, you will not be using file fields directly. As a result, you will need accessor procedures, also commonly referred to as getters and setters. 

Before proceeding, I would like to point out that the example being built in this article uses SQL for all file access. This method can be easily adopted for native file access as well, but requires additional coding. If you are going to use native file access, I recommend you control the open and close status of the file and check in every access procedure for the file status. 

FILE ACCESS PROCEDURES 

You are going to rely on the global properties of your service program to hold file data. The simplest way to do this is to create an externally defined data structure based on your file. I prefer to name the data structure the same as the file and add a prefix. 

h nomain

d mynames       e ds                  extname(MYNAMES) prefix(n_)

  * Prototypes

  /copy qrpglesrc,cp_protos

d PSDS           sds

d  userID               358    367

Note that this code snippet also includes H-specs, the /copy for your prototypes, and a program status data structure. Since we are still in the global section, I'm going to create some additional global variables that may be useful in the future. 

* Global Variables

d sql              s             32767a   varying

d fileName         s                20a   varying inz( 'MYNAMES' )

Now you can begin building your procedures. I'm going to take advantage of RPG IV's long names and prefix all my procedures with the file name. This helps prevent any namespace collisions and increases self-documentation in calling programs. 

The first procedure populates the data structure based on the key field, "ID". This is typically the first procedure called and effectively creates an "instance" of mynames. Once called, the data is available to all the other procedures in the service program. 

 *---------------------------------------------------------------------

 * Select record and populate DS

 *---------------------------------------------------------------------

p mynames_getRecord...

p                 b                   export

d mynames_getRecord...

d                 pi              n

d  id                            5s 0 const

d isFound         s               n   inz(*off)

  /free

    mynames_clear();

  /end-free

c/exec sql

c+ select *

c+   into :mynames

c+   from MYNAMES

c+  where ID = :id

c/end-exec

  /free

    if SQLSTT = '00000' ;

      isFound = *on ;  

    endif ;

    return isFound ;

  /end-free

p mynames_getRecord...

p                 e

The procedure returns an indicator, letting you know whether the selection was successful. This makes it very easy to employ branching logic in the calling code if the record is not found. You'll also notice that the first thing this procedure does is to call another procedure in the same service program, mynames_clear(), which clears the data structure. This is to prevent any residual data problems should the record not be found. This is a useful procedure before inserting records as well, so we'll go ahead and export it: 

 *---------------------------------------------------------------------

 * Clear the mynames DS

 *---------------------------------------------------------------------

p mynames_clear   b                   export

d mynames_clear   pi

  /free

    clear mynames ;

    return ;

  /end-free

p mynames_clear   e

Updating and deleting records are very similar to each other. Below is the procedure for updating a record. 

 *---------------------------------------------------------------------

 * Updating the mynames file

 *---------------------------------------------------------------------

p mynames_update  b                   export

d mynames_update  pi

d isUpdated       s               n   inz(*off)

  /free

    n_UPDUSER = userID ;

    n_UPDSTAMP = %timestamp();

  /end-free

c/exec sql

c+ update mynames

c+    set USERNAME = :n_USERNAME,

c+        USEREMAIL = :n_USEREMAIL,

c+        PHONENUM = :n_PHONENUM,

c+        UPDUSER = :n_UPDUSER,

c+        UPDSTAMP = :n_UPDSTAMP

c+  where ID = :n_ID

c/end-exec

  /free

    if SQLSTT = '00000' ;

      isUpdated = *on ;

    endif ;

    return isUpdated ;

 /end-free

p mynames_update  e

The first thing you have done is made the user name and timestamp information in the file "automatic." By placing this code in the actual update procedure, you are enforcing our rule. Examining this procedure shows you are using the current values in the data structure to update the record. This mimics classic native file access. Adding procedures for deleting and inserting follows the same model. 

FIELD VALUE ACCESS AND USE 

To read and populate the data structure subfields, you need a series of getters and setters. These very small procedures become the most significant because they actually provide the program interface for the field values. From a design standpoint, they could not be simpler. Below is an example of getting and setting the USEREMAIL field value: 

 *---------------------------------------------------------------------

 * Get USEREMAIL

 *---------------------------------------------------------------------

p mynames_getUserEmail...

p                 b                   export

d mynames_getUserEmail...

d                 pi           128a

  /free

    return n_USEREMAIL ;

  /end-free

p mynames_getUserEmail...

p                 e

 *---------------------------------------------------------------------

 * Set USEREMAIL

 *---------------------------------------------------------------------

p mynames_setUserEmail...

p                 b                   export

d mynames_setUserEmail...

d                 pi

d  userEmail                   128a   const

  /free

    n_USEREMAIL = userEmail ;

    return ;

  /end-free 

p mynames_setUserEmail...

p                 e

Include a getter procedure for every field you want to be able to retrieve, and a setter procedure for each one you want to be able to change. A good rule of thumb is to make the return values and the input parameters the same definition as the fields you are accessing. If you examine the complete source code for this article, you will notice that there are no setter procedures for the ID, UPDUSER, or UPDSTAMP fields. That's because I want to control how those values are generated and assigned. 

Looking at the mynames_setEmail() procedure above, if you wanted to institute the e-mail address validation mentioned earlier in this article, this is where you would do so. You could return an indicator, or an error code, or a message. You could easily add logic to the update and insert routines that would not allow the operation to occur if the e-mail wasn't valid. Those implementation details are well beyond the scope of this article, but the concept is clear: you have a significant amount of control inside this one service program to enforce your database rules and maintain data integrity. 

USING THE SERVICE PROGRAM 

You will need to create the prototypes and binder source and compile the *MODULE. Then you can use the *MODULE to create the service program and bind it to a program. Here is a brief example: 

d mynames       e ds                  extname(mynames)

 /copy qrpglesrc,cp_protos

d message         s             50a   inz('User Added.')

  /free

    mynames_clear();

    mynames_setUserName( 'JOEL' );

    mynames_setUserEmail( 'jcochran@itjungle.com');

    if mynames_insert() ;

      dsply message ;

    endif ;

    mynames_getRecord( 1 );

    USERNAME = mynames_getUserName();

    dsply USERNAME ;

    mynames_setUserName( 'RAYMOND' );

    mynames_update();

    USERNAME = mynames_getUserName();

    dsply USERNAME ;

    *inlr = *on ;

  /end-free

FINAL THOUGHTS 

Like any service program, you can include as many procedures as you like. In the example here, you may want to include a procedure for counting the number of records, or returning an array of domain names from the e-mail addresses, or finding the most recently updated record. This design is replete with possibilities. 

Also, be aware that nothing discussed up to this point would prevent someone from sidestepping your service program. This could probably be accomplished by using a trigger program, as could some of the data integrity checking discussed. However a lot of shops are wary of triggers, so this may not be an option for you. The best enforcement is internal policy and programmer discipline. 

I'd like to point out that this is a simple example designed to work for a single record. Many things could be done to improve this program that are beyond the scope of this article. If I were going to put this into production, I would optimize the SQL, use prepared statements, and allow variable keys. I'd also add the capability to handle a group of records using a cursor and a fetchNext() procedure. Finally, I typically add a generic executeSQL procedure, using "execute immediate" as a catch-all for any oddball needs that may arise, as well as a getSQLstate procedure so that you could externally analyze any SQL errors. 

More on RPG Encapsulation 

by Joel Cochran 

Reader response to my July article, "Encapsulating File Access in a Service Program," filled my mailbox for some time. Most of the feedback has been extremely positive, with readers wanting information on some of the more advanced techniques I referred to but didn't have column space to explain. 

Many of the questions had to do with embedded SQL and how to incorporate that into the service program to offer more flexibility. In this article, I address a number of these questions and offer some more techniques to employ in this powerful model. Unlike the basic model I outlined previously, which works easily with either SQL or native I/O, most of these techniques are specific to embedded SQL. Where needed I will explain the SQL aspects of these techniques. 

Using a Template 

I'm a big fan of program templates, such as the subfile templates I've shared in this newsletter before. Well thought out templates allow me to quickly create programs of similar types, and since using the encapsulation model for service programs takes a lot of typing, and if properly implemented would require one service program per file, it is a prime target for a template. For a template to be successful in this regard, we need to add as many standard elements as possible. The idea is to provide an easily modified starting point. 
In this case, I've amended our global variables section to include some fields for our SQL needs: 

// A second DataStructure for SQL statements

d SQL_DS        e ds                  extname(MYNAMES) prefix(sql_)

// The file name.  This was in the previous version but unused.  

// Can be qualified with the library name or made larger to handle 

// SQL table names.

d fileName        s             20a   varying inz( 'MYNAMES' )

// The primary key field name

d keyField        s             10a   varying inz( 'ID' )

// Indicates whether or not the SQL cursors and prepared statements 

// procedure has been executed.

d isSQLprepared   s               n   inz

// Strings for Prepared Statements

d sqlSelectStr    s                   like( sql )

d sqlUpdateStr    s                   like( sql )

d sqlInsertStr    s                   like( sql )

d sqlDeleteStr    s                   like( sql )

Implementing these fields would have little or no affect on a native implementation, but if you implement this model with SQL, you can use these fields in our dynamic SQL statements. This prevents certain efforts from being duplicated and increases flexibility. 

Optimizing the SQL 

One of the potential problems of the basic version of encapsulation is that it uses inline embedded SQL statements. Basically, the SQL engine has to generate, analyze, assign access paths, execute, and destroy each statement every time they execute. Nothing is reused, which adds a lot of unnecessary overhead. To help alleviate that performance burden, we are going to use prepared statements. A prepared statement does all those tasks one time, and then the program holds on to it for reuse. 

To achieve this task, I've added a mynames_prepare() subprocedure. Since this procedure only needs to run once in the service program's lifecycle, I added the isSQLprepared variable above. Now, in the mynames_getRecord(), mynames_update(), mynames_insert(), and mynames_delete() procedures, check this variable first and run the mynames_prepare() procedure if it is false: 

/free

  if not isSQLprepared ;

    mynames_prepare();

  endif ;

/end-free

Examine the mynames_prepare() subprocedure in the provided source code, and you'll see a couple of interesting things. First, if you've never used prepared statements before, this will give you an easy template to copy. Second, speaking of templates, you'll notice that all the statement strings use the filename and keyField variables where applicable, instead of hard-coding them into the strings themselves. This goes back to templating: changing the initial value of a variable in a D-spec is much easier to maintain and makes this approach easy to copy for other applications. You'll also notice that the procedure returns an indicator for whether any of the statements encountered a problem, but permits the rest of the statements to continue. Since this service program and the file it serves should constantly remain in synch, this is primarily used for debug purposes. 

Implementing the Prepared Statements 

With the prepared statements in the bag, we need to implement them in the database access procedures. The update, insert, and delete statements are executed with their relative prepared statements. The question marks (?) in the strings for the prepared statements act as variable place holders, so when we execute the statement we add a list of variables using the "Using" option. Below is the altered mynames_update() procedure: 

p mynames_update  b                   export

d mynames_update  pi              n

d isUpdated       s               n   inz(*off)

 /free

   if not isSQLprepared ;

     mynames_prepare();

   endif ;

   n_UPDUSER = userID ;

   n_UPDSTAMP = %timestamp();

 /end-free

c/exec sql

c+ execute updateFile

c+   using :n_USERNAME,

c+         :n_USEREMAIL,

c+         :n_PHONENUM,

c+         :n_UPDUSER,

c+         :n_UPDSTAMP,

c+         :n_ID

c/end-exec

 /free

   if SQLSTT = '00000' ;

     isUpdated = *on ;

   endif ;

   return isUpdated ;

 /end-free

p mynames_update  e

Follow this same model for the insert and delete procedures. You may also want to read up on SQL States in the "SQL Programming Concepts" manual, coverage of which is beyond the scope of this article. 

Dealing with Multiple Record Sets 

The encapsulation model from the last article is intended to emulate an object-oriented approach, where the object holds and serves data from a single database record. Frequently, though, what we really want is to work with a subset of records: more than one, but not the whole file. In other languages, this is referred to as a record set. The flexibility of SQL allows us to process these subsets dynamically! This can also be achieved with OPNQRYF, but otherwise cannot be done with native I/O without a ton of logical files and a huge branch statement. 

To process a record set, we are going to create three new procedures. First, using the techniques from the mynames_prepare() procedure, we will create the mynames_getRecordSet() procedure. This procedure receives two parms, one for the where clause and an optional one for order by: 

p mynames_getRecordSet...

p                 b                   export

d mynames_getRecordSet...

d                 pi              n

d  whereClause                1024a   const varying

d  orderByClause              1024a   const varying options(*nopass)

d where           s           1024a   varying inz( ' where ' )

d orderBy         s           1024a   varying inz( ' order by ' )

d isOpen          s               n   inz(*off)

 /free

   where = where + whereClause ;

   if %parms() = 1 ;

     orderBy = orderBy + keyField ;

   else ;

     orderBy = orderBy + orderByClause ;

   endif ;

   sqlRecSetStr = 'select * from ' + fileName + where + orderBy ;

   mynames_closeRecordSet();

 /end-free

c/exec sql

c+ declare setCursor cursor

c+     for setStatement

c/end-exec

c/exec sql

c+ prepare setStatement

c+    from :sqlRecSetStr

c/end-exec

c/exec sql

c+ open setCursor

c/end-exec

/free

   if sqlstt = '00000' ;

     isOpen = *on ;

   endif ;

   return isOpen ;

 /end-free

p mynames_getRecordSet...

p                 e

Before executing any of the SQL statements, we call another new procedure, mynames_closeRecordSet(). This ensures that the cursor is closed before we try to open it based on our new prepared statement: 

p mynames_closeRecordSet...

p                 b                   export

d mynames_closeRecordSet...

d                 pi

c/exec sql

c+ close setCursor

c/end-exec

 /free

   return ;

 /end-free

p mynames_closeRecordSet...

p                 e

Now that we have a method for opening a record set, we need a procedure for reading the next record in the set: 

p mynames_fetchRecordSet...

p                 b                   export

d mynames_fetchRecordSet...

d                 pi              n

d isFound         s               n   inz( *off )

c/exec sql

c+ fetch next

c+  from setCursor

c+  into :mynames

c/end-exec

 /free

   if sqlstt = '00000' ;

     isFound = *on ;

   endif ;

   return isFound ;

 /end-free

p mynames_fetchRecordSet...

p                 e

The procedure returns *off when there are no more records to process. Here is a brief example of using and processing a record set: 

    // Retrieves the record set of all records with

    // PHONENUM equal to 8005551212 and orders the

    // results by USERNAME

    // returns true if SQL statement is OK and set is opened

    if mynames_getRecordSet( 'PHONENUM = 8005551212' : 'USERNAME' );

      // Do While loop for records remaining

      dow mynames_fetchRecordSet();

        message = 'User: ' + %trimr( mynames_getUserName() ) +

                 ' EMail: ' + %trimr( mynames_getUserEmail() );

        dsply message ;

      enddo ;

    endif ;

This record set approach could also be enhanced by making the cursor "scrollable," allowing you to jump around within the record set. Of course, you would need to implement additional procedures such as mynames_fetchFirstRecord(), mynames_fetchPreviousRecord(), and mynames_fetchLastRecord(). 

Additional Procedures 

Hopefully by now you've come to see that with this approach you simply add procedures for whatever you need. To handle alternative or multiple keys, add procedures that access the database in the necessary manner. Just make sure that the procedures populate the main data structure, and the rest of the procedures will continue to function as desired. You can also easily add procedures that validate data entry, like validating if a particular name or ID is on file, or if an e-mail address is properly formatted. Need to find the average value of a particular field? Add a procedure. How about one that deletes records based on date? Add a procedure. Want to know the standard deviation of the sum of three fields? You guessed it: add a procedure! The more procedures you add, the more functional the service program. 

A Final SQL Catch-All Procedure 

As a final backstop for unanticipated file maintenance, you can add a simple _executeSQL() procedure. The parameters for this procedure include the statement type, the values string, and the where statement: 

*---------------------------------------------------------------------

 * Execute dynamic SQL statement

 *---------------------------------------------------------------------

p mynames_executeDynamicSQL...

p                 b                   export

d mynames_executeDynamicSQL...

d                 pi             5a

d  type                          6a   const

d  values                     1024a   const varying options(*omit)

d  where                      1024a   const varying options(*nopass)

d isOK            s               n   inz( *off )

 /free

   if type = 'UPDATE' and %parms() = 3 ;

     executeStr = 'update ' + fileName +

                  ' set ' + values +

                  ' where ' + where ;

     isOK = *on ;

   elseif type = 'UPDATE' and %parms() = 2 ;

     executeStr = 'update ' + fileName +

                  ' set ' + values ;

     isOK = *on ;

   elseif type = 'DELETE' and %parms() = 1 ;

     executeStr = 'delete from ' + fileName ;

     isOK = *on ;

   elseif type = 'DELETE' and %parms() = 3 ;

     executeStr = 'delete from ' + fileName +

                  ' where ' + where ;

     isOK = *on ;

   elseif type = 'INSERT' and %parms()>= 2 and

          %addr( values ) <> *null ;

     executeStr = 'insert into ' + fileName +

                  ' ' + values ;

     isOK = *on ;

   else ;

     sqlStt = '99999' ;

   endif ;

   if isOK ;

 /end-free

c/exec sql

c+   execute immediate :executeStr

c/end-exec

 /free

   endif ;

   return sqlstt ;

 /end-free

p mynames_executeDynamicSQL...

p                 e

Right away, though, there is a problem with this approach: it opens a serious control gap. This procedure could easily be exploited to run a statement that bypasses controls specifically included in the other procedures. In other words, it seriously contradicts the entire purpose of encapsulation, so think hard before adding this procedure to your service program. 

You might wonder why I even bother to mention it: first of all, if I know the service program is just going to be used internally, or more specifically just by me, then I don't worry about another programmer maliciously using it. Second, it allows me to solve a one-off problem quickly without needing to add a new procedure that I'll never use again. It basically allows me to be lazy. As a vendor, I would never use this in a situation where someone else could abuse the procedure: abusing my code is a pleasure I reserve for myself. 

Embedding SQL in /free 

by Joel Cochran 

I've been hooked on free-format RPG since I first upgraded to V5R1. In fact, I have immersed myself so deeply into /free that I almost feel uncomfortable when I have to maintain fixed-format code. I'll assume if you are reading this article that you are already aware of the virtues of free-format. In total fairness, I will also acknowledge that working with free-format certainly has some undesirable aspects. Listing and discussing them would be an entire article in its own right, and maybe one I'll contemplate soon. But in this article I'd like to address the one that personally bugs me the most: embedding SQL in free-format code. 

A Quick How-To 

Regular readers of Four Hundred Guru already know I'm a big fan of SQL. To embed SQL into an RPG program, the SQL pre-compiler has to go through the source code first and convert the SQL statements to program calls. To do this, we have to signal the pre-compiler that a particular section of source code is in fact SQL by wrapping the code in c/exec sql and c/end-exec compiler directives. On top of that, each line inside these directives must have c+ in positions 6 and 7. Here is an example: 

c/exec sql

c+   select *

c+     into :mainDS

c+     from MYLIB/MYFILE

c+    where IDFIELD = :KEY_VALUE

c/end-exec
For the embedded SQL newcomer, the statement above will get all the fields from a record and place the contents into a data structure named mainDS. This block of code would execute just fine in either fixed- or free-format, but free-format it means first issuing a /end-free statement and then reissueing a /free statement after the SQL. There is lots of room for improvement here, and more than likely we want to deal with a group of records, which means DECLARE CURSOR, OPEN, FETCH, CLOSE, and possibly PREPARE statements. Explanation of these is beyond the scope of this article, but here is a sample bit of code that uses all the above in free-format: 

d mainDS        e ds                  extname(MYNAMES)

d sql             s          32000a   varying

d                                     inz( 'Select * from mynames +

d                                     order by USERNAME' )

d message         s             52a   inz

c/exec sql

c+ declare mainCursor cursor

c+     for mainStatement

c/end-exec

c/exec sql

c+ prepare mainStatement

c+    from :sql

c/end-exec

c/exec sql

c+ open mainCursor

c/end-exec

c/exec sql

c+ fetch next from mainCursor

c+  into :mainDS

c/end-exec

 /free

   dow sqlstt = '00000' ;

     message = %trimr( m_USERNAME ) + ' ' + %trimr( m_USEREMAIL ); 

     dsply message ;

 /end-free

c/exec sql

c+   fetch next from mainCursor

c+    into :mainDS

c/end-exec

 /free

    enddo ;

 /end-free

c/exec sql

c+ close mainCursor

c/end-exec

 /free

    *inlr = *on ;

 /end-free

The Ugly Factor 

The issue with embedded SQL in free-format has primarily to do with the need to constantly drop in and out of /free to execute SQL statements. As demonstrated above, this can be quite messy. In fact, this is downright ugly. And it can easily get worse! Multiple cursors, additional select statements, updates, deletes, and inserts can all be interspersed with sections of free-format code. Eventually you have as much time invested in compiler directives as you do in actual code. This approach can be difficult to read, which means it is difficult to manage. 

I have personally written enough code with embedded SQL that I can pretty much ignore the ugliness, but many people contend that we shouldn't have to deal with this in the first place. There is enough discontent with this particular topic that some people refuse to use /free. IBM should be paying attention: If the party line is to adopt free-format RPG, then it needs to be cleaner, especially when dealing with SQL. No matter your opinion on the matter--and there are many--this is the current state of things, so let's discuss getting rid of the ugliness. 

So how do we conquer the ugliness? Embedded SQL requires the c/exec sql and c/end-exec blocks, so there isn't any way to get rid of those. And we already know that we cannot use embedded SQL statements inside free-format. Our only recourse then is to separate the embedded SQL from the free-format code. Naturally, for such an operation, we are going to turn to our trusty tool: the subprocedure. 
Using SQL in Subprocedures 

Subprocedures are an excellent mechanism for detaching our ugly SQL statements from our program logic. Ultimately, since we can't change the rules, the ugliness still remains, but the final affect is much more palatable. Before we begin breaking our program up into smaller chunks, there are some pesky rules to go over. 

First, the SQL Precompiler, at least through V5R2, requires that some of the SQL statements occur in the source code in the correct physical order. Specifically, since the DECLARE statement must be logically executed before the related OPEN the precompiler expects them to be in this order in the actual source code. If you try to compile a program that has the OPEN statement physically before the DECLARE statement, you should receive an SQL0504 ("Position xx Cursor %CURSORNAME% not declared.") compile error. The same rule however is not true for OPEN and FETCH statements, which compile just fine out of order. Honestly, I don't know if this applies to any others or not, but to avoid confusion and compile problems, I always try to place the subprocedures in the same physical order as their logical execution. 

Second, the use of local variables. Since we are moving these statements to subprocedures, we can take advantage of local variables, but with restrictions. Namely, if the variable name is also not a global variable name, an SQL statement should be able to use the local variable as a host variable. To complicate matters, this is not always true. The local name can be the same as a global name, provided that both variables are defined the same. Otherwise, SQL will get confused and not recognize the value of either variable. This is only a problem if the local and global variables are defined differently. But don't let any of this odd behavior fool you into thinking something is wrong with the SQL precompiler: the manual in V5R3 still states: "The names of host variables must be unique within the program, even if the host variables are in different procedures." So ultimately, this has nothing to do with global versus local! The solution is to ensure that your host variable names unique no matter what. 

To avoid confusion, I've adopted the convention that all host variables should be global. For update and insert statements, I like to use an externally defined DS based on the same file. This way the variable definitions are guaranteed to be correct. As a result, no data conversion is ever required so I get a little performance boost. You can populate the variables before calling the procedure or use local variables to populate the global host variables if you like. While it is a little extra work, being cognizant of this issue beforehand can save you time and aggravation. 

Modularizing the Program 

Now that we know what the rules are, we can transfer the SQL statements out of our mainline and into subprocedures. This really couldn't be any easier: Simply move the offending statements to a basic procedure shell, add the prototype to the D-Specs, and replace the original code with the new procedure call. Apply this to the code posted above and here is the result: 

d mainDS        e ds                  extname(MYNAMES) prefix(m_)

 * Prototypes

d declare         pr

d openCursor      pr

d fetchNext       pr

d closeCursor     pr

 * Standalone Variables

d sql             s          32000a   varying

d                                     inz( 'Select * from mynames +

d                                     order by USERNAME' )

d message         s             52a   inz

 /free

   declare();

   openCursor();

   fetchNext();

   dow sqlstt = '00000' ;

     message = %trimr( m_USERNAME ) + ' ' + %trimr( m_USEREMAIL ); 

     dsply message ;

     fetchNext();

   enddo ;

   closeCursor();

   *inlr = *on ;

 /end-free

 *-------------------------------------------------------------------

p declare         b

d declare         pi

c/exec sql

c+ declare mainCursor cursor

c+     for mainStatement

c/end-exec

c/exec sql

c+ prepare mainStatement

c+    from :sql

c/end-exec

p declare         e

 *-------------------------------------------------------------------

p openCursor      b

d openCursor      pi

c/exec sql

c+ open mainCursor  

c/end-exec

p openCursor      e

 *-------------------------------------------------------------------

p fetchNext       b

d fetchNext       pi

c/exec sql

c+ fetch next from mainCursor

c+  into :mainDS   

c/end-exec

p fetchNext       e 

 *-------------------------------------------------------------------

p closeCursor     b

d closeCursor     pi

c/exec sql

c+ close mainCursor

c/end-exec

p closeCursor     e

 *-------------------------------------------------------------------

You'll notice quickly that there isn't much to any of the procedures. The real gem here is in the main line. Hopefully you'll agree that this is much more palatable than what we started with. We certainly have a lot less jumping in and out of free-format, even in the subprocedures, since they are primarily SQL. 

Obviously this is a very simple example, but let me point out a couple of things. At first glance, it appears that the declare() and open() procedures could be combined. While in this case that may be true, what if the cursor had a variable where clause? To handle this, the SQL statement would include a parameter marker for the where value: 

d sql             s          32000a   varying

d                                     inz( 'Select * from mynames +

d                                     where USEREMAIL like (?) +

d                                     order by USERNAME' )

If the declare() and open() procedures were combined, then the statement would get re-prepared and the cursor re-declared every time the where clause changes. That's a lot of unnecessary overhead, so most of the time it makes sense to separate these into their own procedures. If you really want to get the declare() statement out of the mainline, try setting a global variable such as isDeclared and checking its state in the open() subprocedure, calling declare() if it the subprocedure hasn't been executed yet.

A Few More Tidbits 

For scrollable cursors, is to have a single fetch procedure (instead of our unidirectional fetchNext) that accepts a parameter for the fetch method and some branching logic. Below is a simple example, you could easily add the rest of the scroll types (i.e. FIRST, LAST, CURRENT, etc.) to this code. 

In the declare() subprocedure: 
c/exec sql

c+ declare mainCursor scroll cursor

c+     for mainStatement

c/end-exec

And the fetch() subprocedure: 

p fetch           b

d fetch           pi

d  scrollType                   10a   const options(*nopass)

 /free

   if %parms() = 0 or scrollType = 'NEXT' ;

 /end-free

c/exec sql

c+   fetch next from mainCursor

c+    into :mainDS  

c/end-exec

 /free

   elseif scrollType = 'PRIOR' ;

 /end-free

c/exec sql

c+   fetch prior from mainCursor

c+    into :mainDS

c/end-exec

 /free

   endif ;

 /end-free

p fetch           e

Granted, a bit of the ugliness has returned with this sort of branching. Remember, I said we couldn't get rid of the ugliness, but we could reduce it. At least we are limiting it to this subprocedure and getting it out of the mainline. And of course, you could have separate subprocedures for each of these instead of one subprocedure. 

Another thing you could do is to make your fetch() routine(s) return a boolean value representing whether or not a record was found. I'm not going to demonstrate it, just check the value of SQLSTT and return *ON or *OFF based on your particular needs. The benefit of this is a little less mainline: 

/free

   declare();

   openCursor();

   dow fetchNext();

     message = %trimr( m_USERNAME ) + ' ' + %trimr( m_USEREMAIL ); 

     dsply message ;

   enddo ;

   closeCursor();

   *inlr = *on ;

 /end-free

You also no longer have the "priming" call to fetchNext() and the "continuing" call. 

Final Thoughts 

I do, as always, have a few items on my wish list. The first is to find a solution to the problems with the local variables discussed above. Second, I'd like to have enhanced cursor support, more like an actual OO-style result set. Looping through a cursor is OK, but sometimes I want to know how many records the cursor has, or maybe dump the contents of a cursor into an array or even an outfile. I can imagine a lot of similar tasks and uses. I'd really like to be able to pass a cursor between programs and to be able to retrieve data from a cursor by referencing field name or ordinal position. Mostly, and more in line with the topic of this article, I'd like to see a set of SQL Built-In Functions (BIFs). For example, %execSQL(), %declareSQL(), %prepareSQL(), %opencursor(), %fetchSQL(), etc. could certainly eliminate the need for the c/exec-sql and c/end-exec directives, which would go a long way to improving the use of SQL in free-format RPG. 

In the meantime, while the techniques in this article aren't perfect, they can simplify and cleanse your mainline code. This is especially true when you have a lot of conditional updating, multiple cursors, and the like. It certainly provides some relief for those of us wishing to take advantage of both SQL and free-format RPG.

